

# AVL Trees

D Oisín Kidney

July 28, 2018

## Abstract

This is a verified implementation of AVL trees in Agda, taking ideas primarily from Conor McBride’s paper “How to Keep Your Neighbours in Order” [2] and the Agda standard library [1].

## Contents

|          |                                      |          |
|----------|--------------------------------------|----------|
| <b>1</b> | <b>Introduction</b>                  | <b>1</b> |
| <b>2</b> | <b>Bounded</b>                       | <b>2</b> |
| <b>3</b> | <b>Balance</b>                       | <b>3</b> |
| <b>4</b> | <b>The Tree Type</b>                 | <b>3</b> |
| <b>5</b> | <b>Rotations</b>                     | <b>4</b> |
| 5.1      | Right Rotation . . . . .             | 4        |
| 5.2      | Left Rotation . . . . .              | 5        |
| <b>6</b> | <b>Insertion</b>                     | <b>6</b> |
| <b>7</b> | <b>Lookup</b>                        | <b>6</b> |
| <b>8</b> | <b>Deletion</b>                      | <b>7</b> |
| 8.1      | Uncons . . . . .                     | 7        |
| 8.2      | Widening . . . . .                   | 8        |
| 8.3      | Full Deletion . . . . .              | 9        |
| <b>9</b> | <b>Packaging</b>                     | <b>9</b> |
| 9.1      | Dependent Map . . . . .              | 10       |
| 9.2      | Non-Dependent (Simple) Map . . . . . | 10       |
| 9.3      | Set . . . . .                        | 11       |

## 1 Introduction

First, some imports.

```

{-# OPTIONS --without-K #-}

open import Relation.Binary
open import Relation.Binary.PropositionalEquality
open import Level using (Lift; lift;  $\sqcup$ _; lower)
open import Data.Nat as  $\mathbb{N}$  using ( $\mathbb{N}$ ; suc; zero; pred)
open import Data.Product
open import Data.Unit renaming ( $\top$  to 1)
open import Data.Maybe
open import Function
open import Data.Bool
open import Data.Empty renaming ( $\perp$  to 0)

```

Next, We declare a module: the entirety of the following code is parameterized over the *key* type, and a strict total order on that key.

```

module AVL
  {k r} (Key : Set k)
  {_ <_ : Rel Key r}
  (isStrictTotalOrder : IsStrictTotalOrder  $\equiv$  _ <_ )
  where

  open IsStrictTotalOrder isStrictTotalOrder

```

## 2 Bounded

The basic idea of the verified implementation is to store in each leaf a proof that the upper and lower bounds of the trees to its left and right are ordered appropriately.

Accordingly, the tree type itself will have to have the upper and lower bounds in its indices. But what are the upper and lower bounds of a tree with no neighbours? To describe this case, we add lower and upper bounds to our key type.

```

module Bounded where

infix 5  $\sqcup$ 

data  $\sqcup$  : Set k where
   $\perp$   $\top$  :  $\sqcup$ 
   $\sqcup$  : (k : Key)  $\rightarrow$   $\sqcup$ 

```

This type itself admits an ordering relation.

```

infix 4  $\sqcup$  <_

 $\sqcup$  <_ :  $\sqcup \rightarrow \sqcup \rightarrow$  Set r
 $\perp$  <  $\perp$  = Lift r 0
 $\perp$  <  $\top$  = Lift r 1
 $\perp$  <  $\sqcup$  = Lift r 1

```

$$\begin{aligned}
\top & \quad \top < \_ &= \text{Lift } r \ 0 \\
[\_] & \quad \top < \_ &= \text{Lift } r \ 0 \\
[\_] & \quad \top < \top &= \text{Lift } r \ 1 \\
[x] & \quad \top < [y] &= x < y
\end{aligned}$$

Finally, we can describe a value as being “in bounds” like so.

$$\begin{aligned}
& \text{infix 4 } \_ < \_ < \_ \\
& \_ < \_ < \_ : \top \rightarrow \text{Key} \rightarrow \top \rightarrow \text{Set } r \\
& l < x < u = l \top < [x] \times [x] \top < u
\end{aligned}$$

### 3 Balance

To describe the balance of the tree, we use the following type:

$$\begin{aligned}
& \text{data } (\_ \sqcup \_) \equiv \_ : \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N} \rightarrow \text{Set where} \\
& \swarrow : \forall \{n\} \rightarrow (\text{suc } n \sqcup \_ \equiv \text{suc } n) \equiv \text{suc } n \\
& \dashv : \forall \{n\} \rightarrow (\_ \sqcup \_ \equiv \_ \equiv n) \\
& \searrow : \forall \{n\} \rightarrow (\_ \sqcup \text{suc } n \equiv \text{suc } n)
\end{aligned}$$

The tree can be either left- or right-heavy (by one), or even. The indices of the type are phrased as a proof:

$$\max(x, y) = z \quad (1)$$

The height of a tree is the maximum height of its two subtrees, plus one. Storing a proof of the maximum in this way will prove useful later.

We will also need some combinators for balance:

$$\begin{aligned}
& \swarrow \Rightarrow \swarrow : \forall \{x \ y \ z\} \rightarrow (x \sqcup y) \equiv z \rightarrow (z \sqcup x) \equiv z \\
& \swarrow \Rightarrow \swarrow \swarrow = \dashv \\
& \swarrow \Rightarrow \swarrow \dashv = \dashv \\
& \swarrow \Rightarrow \swarrow \searrow = \swarrow \\
& \swarrow \Rightarrow \searrow : \forall \{x \ y \ z\} \rightarrow (x \sqcup y) \equiv z \rightarrow (y \sqcup z) \equiv z \\
& \swarrow \Rightarrow \searrow \swarrow = \searrow \\
& \swarrow \Rightarrow \searrow \dashv = \dashv \\
& \swarrow \Rightarrow \searrow \searrow = \dashv
\end{aligned}$$

### 4 The Tree Type

The type itself is indexed by the lower and upper bounds, some value to store with the keys, and a height. In using the balance type defined earlier, we ensure that the children of a node cannot differ in height by more than 1. The bounds proofs also ensure that the tree must be ordered correctly.

$$\begin{aligned}
& \text{data Tree } \{v\} (V : \text{Key} \rightarrow \text{Set } v) (l \ u : \top) : \mathbb{N} \rightarrow \text{Set } (k \sqcup v \sqcup r) \text{ where} \\
& \text{leaf} : (l < u : l \top < u) \rightarrow \text{Tree } V \ l \ u \ 0
\end{aligned}$$

```

node :  $\forall \{h \text{ } lh \text{ } rh\}$ 
      (k : Key)
      (v : V k)
      (bl :  $\langle lh \sqcup rh \rangle \equiv h$ )
      (lk : Tree V l [ k ] lh)
      (ku : Tree V [ k ] u rh)  $\rightarrow$ 
      Tree V l u (suc h)

```

## 5 Rotations

AVL trees are rebalanced by rotations: if, after an insert or deletion, the balance invariant has been violated, one of these rotations is performed as correction.

Before we implement the rotations, we need a type to describe a tree whose height may have changed:

```

Inserted :  $\forall \{v\} (V : Key \rightarrow Set \ v) (l \ u : \mathbb{I}) (n : \mathbb{N}) \rightarrow Set \ (k \sqcup v \sqcup r)$ 
Inserted V l u n =  $\exists [inc?] ] Tree \ V \ l \ u \ (if \ inc? \ then \ suc \ n \ else \ n)$ 
pattern 0+ tr = false , tr
pattern 1+ tr = true  , tr

```

### 5.1 Right Rotation

When the left subtree becomes too heavy, we rotate the tree to the right.

```

rotr :  $\forall \{lb \ ub \ rh \ v\} \{V : Key \rightarrow Set \ v\}$ 
       $\rightarrow (k : Key)$ 
       $\rightarrow V \ k$ 
       $\rightarrow Tree \ V \ lb \ [ \ k \ ] \ (suc \ (suc \ rh))$ 
       $\rightarrow Tree \ V \ [ \ k \ ] \ ub \ rh$ 
       $\rightarrow Inserted \ V \ lb \ ub \ (suc \ (suc \ rh))$ 

```

This rotation comes in two varieties: single and double. Single rotation can be seen in figure 1.




Figure 1: Single right-rotation

```

rotr x xv (node y yv < a b) c = 0+ (node y yv = a (node x xv = b c))
rotr x xv (node y yv = a b) c = 1+ (node y yv > a (node x xv < b c))

```

And double rotation in figure 2.




Figure 2: Double right-rotation

$$\begin{aligned} \text{rot}^r \ x \ xv \ (\text{node } y \ yv \ \backslash \ a \ (\text{node } z \ zv \ bl \ b \ c)) \ d = \\ 0+ \ (\text{node } z \ zv \ \vdash \ (\text{node } y \ yv \ (\backslash \Rightarrow \swarrow \ bl) \ a \ b) \ (\text{node } x \ xv \ (\swarrow \Rightarrow \backslash \ bl) \ c \ d)) \end{aligned}$$

## 5.2 Left Rotation

Left-rotation is essentially the inverse of right.

$$\begin{aligned} \text{rot}^l : \forall \{lb \ ub \ lh \ v\} \{V : Key \rightarrow \text{Set } v\} \\ \rightarrow (k : Key) \\ \rightarrow V \ k \\ \rightarrow \text{Tree } V \ lb \ [k] \ lh \\ \rightarrow \text{Tree } V \ [k] \ ub \ (\text{suc } (\text{suc } lh)) \\ \rightarrow \text{Inserted } V \ lb \ ub \ (\text{suc } (\text{suc } lh)) \end{aligned}$$

Single:




Figure 3: Single left-rotation

$$\begin{aligned} \text{rot}^l \ x \ xv \ c \ (\text{node } y \ yv \ \swarrow \ b \ a) &= 0+ \ (\text{node } y \ yv \ \vdash \ (\text{node } x \ xv \ \vdash \ c \ b) \ a) \\ \text{rot}^l \ x \ xv \ c \ (\text{node } y \ yv \ \vdash \ b \ a) &= 1+ \ (\text{node } y \ yv \ \swarrow \ (\text{node } x \ xv \ \swarrow \ c \ b) \ a) \end{aligned}$$

and double:

$$\begin{aligned} \text{rot}^l \ x \ xv \ d \ (\text{node } y \ yv \ \swarrow \ (\text{node } z \ zv \ bl \ c \ b) \ a) = \\ 0+ \ (\text{node } z \ zv \ \vdash \ (\text{node } x \ xv \ (\backslash \Rightarrow \swarrow \ bl) \ d \ c) \ (\text{node } y \ yv \ (\swarrow \Rightarrow \backslash \ bl) \ b \ a)) \end{aligned}$$




Figure 4: Double left-rotation

## 6 Insertion

After the rotations, insertion is relatively easy. We allow the caller to supply a combining function.

```

insert : ∀ {l u h v} {V : Key → Set v} (k : Key)
  → V k
  → (V k → V k → V k)
  → Tree V l u h
  → l < k < u
  → Inserted V l u h
insert v vc f (leaf l < u) (l, u) = 1+ (node v vc ÷ (leaf l) (leaf u))
insert v vc f (node k kc bl tl tr) prf with compare v k
insert v vc f (node k kc bl tl tr) (l, _)
  | tri < a _ _ with insert v vc f tl (l, a)
... | 0+ tl' = 0+ (node k kc bl tl' tr)
... | 1+ tl' with bl
...   | ↗ = rotr k kc tl' tr
...   | ÷ = 1+ (node k kc ↗ tl' tr)
...   | ↘ = 0+ (node k kc ÷ tl' tr)
insert v vc f (node k kc bl tl tr) _
  | tri ≈ _ refl _ = 0+ (node k (f vc kc) bl tl tr)
insert v vc f (node k kc bl tl tr) (_, u)
  | tri > _ _ c with insert v vc f tr (c, u)
... | 0+ tr' = 0+ (node k kc bl tl tr')
... | 1+ tr' with bl
...   | ↗ = 0+ (node k kc ÷ tl tr')
...   | ÷ = 1+ (node k kc ↘ tl tr')
...   | ↘ = rotl k kc tl tr'

```

## 7 Lookup

Lookup is also very simple. No invariants are needed here.

```

lookup : (k : Key)
  → ∀ {l u s v} {V : Key → Set v}
  → Tree V l u s

```

```

→ Maybe (V k)
lookup k (leaf l < u) = nothing
lookup k (node v vc _ tl tr) with compare k v
... | tri < _ _ _ = lookup k tl
... | tri ≈ _ refl _ = just vc
... | tri > _ _ _ = lookup k tr

```

## 8 Deletion

Deletion is by far the most complex operation out of the three provided here. For deletion from a normal BST, you go to the node where the desired value is, perform an “uncons” operation on the right subtree, and use that to rebuild and rebalance the tree.

### 8.1 Uncons

First then, we need to define “uncons”. We’ll use a custom type as the return type from our uncons function, which stores the minimum element from the tree, and the rest of the tree:

```

data Deleted {v} (V : Key → Set v) (lb ub : ℤ) : ℕ → Set (k ⊔ v ⊔ r) where
  -0 : ∀ {n} → Tree V lb ub n → Deleted V lb ub n
  -1 : ∀ {n} → Tree V lb ub n → Deleted V lb ub (suc n)

deleted : ∀ {v} {V : Key → Set v} {lb ub n}
  → Inserted V lb ub n
  → Deleted V lb ub (suc n)
deleted (0+ x) = x -1
deleted (1+ x) = x -0

record Cons {v}
  (V : Key → Set v)
  (lb ub : ℤ)
  (h : ℕ) : Set (k ⊔ v ⊔ r) where
  constructor cons
  field
    head : Key
    val   : V head
    l < u : lb ℤ < [ head ]
    tail  : Deleted V [ head ] ub h

```

You’ll notice it also stores a proof that the extracted element preserves the lower bound.

The uncons function itself is written in a continuation-passing style.

```

uncons : ∀ {lb ub h lh rh v} {V : Key → Set v}
  → (k : Key)
  → V k
  → ⟨ lh ⊔ rh ⟩ ≡ h

```

```

→ Tree V lb [ k ] lh
→ Tree V [ k ] ub rh
→ Cons V lb ub (suc h)
uncons k v bl tl tr = go k v bl tl tr id
where
go : ∀ {lb ub h lh rh v ub' h'} {V : Key → Set v}
→ (k : Key)
→ V k
→ (lh ⊔ rh) ≡ h
→ Tree V lb [ k ] lh
→ Tree V [ k ] ub rh
→ (∀ {lb'} → Deleted V [ lb' ] ub (suc h) → Deleted V [ lb' ] ub' (suc h'))
→ Cons V lb ub' (suc h')
go k v = (leaf l<u) tr c = cons k v l<u (c (tr -1))
go k v = (node kl vl bl tl trl) tr c = go kl vl bl tl trl
λ { (tl' -1) → c ((node k v ∖ tl' tr) -0)
; (tl' -0) → c ((node k v ∖ tl' tr) -0) }
go k v ∖ (leaf l<u) tr c = cons k v l<u (c (tr -1))
go k v ∖ (node kl vl bl tl trl) tr c = go kl vl bl tl trl
λ { (tl' -1) → c (deleted (rotl k v tl' tr))
; (tl' -0) → c ((node k v ∖ tl' tr) -0) }
go k v ∖ (node kl vl bl tl trl) tr c = go kl vl bl tl trl
λ { (tl' -1) → c ((node k v ∖ tl' tr) -1)
; (tl' -0) → c ((node k v ∖ tl' tr) -0) }

```

## 8.2 Widening

To join the two subtrees together after a deletion operation, we need to weaken (or widen) the bounds of the left tree. This is an  $\mathcal{O}(\log n)$  operation.

For the widening, we'll need some properties on orderings:

```

x<⊥ : ∀ {x} → x ⊥ < ⊥ → Lift r 0
x<⊥ {⊥} = lift o lower
x<⊥ {⊤} = lift o lower
x<⊥ {[ _ ]} = lift o lower

⊥<-trans : ∀ x {y z} → x ⊥ < y → y ⊥ < z → x ⊥ < z
⊥<-trans ⊥ {y} {⊥} _ y<z = x<⊥ {x = y} y<z
⊥<-trans ⊥ {⊥} {⊤} _ _ = _
⊥<-trans ⊥ {⊥} {[ _ ]} _ _ = _
⊥<-trans ⊤ {⊥} {⊥} (lift ()) _ = _
⊥<-trans [ _ ] {y} {⊥} _ y<z = x<⊥ {x = y} y<z
⊥<-trans [ _ ] {⊥} {⊤} _ = _
⊥<-trans [ _ ] {⊥} {[ _ ]} (lift ()) _ = _
⊥<-trans [ _ ] {⊤} {[ _ ]} (lift ()) _ = _
⊥<-trans [ x ] {[ y ]} {[ z ]} x<y y<z =
IsStrictTotalOrder.trans isStrictTotalOrder x<y y<z

```



Finally, the widen function itself simply walks down the right branch of the tree until it hits a leaf.

```

widen : ∀ {lb ub ub' h v} {V : Key → Set v}
  → ub ⊔ < ub'
  → Tree V lb ub h
  → Tree V lb ub' h
widen {lb} ub < ub' (leaf l < u) = leaf (⊔ <-trans lb l < u ub < ub')
widen ub < ub' (node k v bl tl tr) = node k v bl tl (widen ub < ub' tr)

```

### 8.3 Full Deletion

The deletion function is by no means simple, but it does maintain the correct complexity bounds.

```

delete : ∀ {lb ub h v} {V : Key → Set v}
  → (k : Key)
  → Tree V lb ub h
  → Deleted V lb ub h
delete _ (leaf l < u) = leaf l < u -0
delete k (node k1 v b tl tr) with compare k k1
delete k (node k1 v b tl tr) | tri < _ _ _ with delete k tl | b
... | tl' -1 | < = node k1 v b tl' tr -1
... | tl' -1 | = = node k1 v b tl' tr -0
... | tl' -1 | > = deleted (rotl k1 v tl' tr)
... | tl' -0 | _ = node k1 v b tl' tr -0
delete {lb} k (node k v b tl (leaf k < ub)) | tri ≈ _ refl _ with b | tl
... | < | _ = widen k < ub tl -1
... | = | leaf lb < k = leaf (⊔ <-trans lb lb < k k < ub) -1
delete k (node k v b tl (node kr vr br tlr trr})) | tri ≈ _ refl _
  with b | uncons kr vr br tlr trr
... | < | cons k' v' l < u (tr' -1) = deleted (rotr k' v' (widen l < u tl) tr')
... | < | cons k' v' l < u (tr' -0) = node k' v' < (widen l < u tl) tr' -0
... | = | cons k' v' l < u (tr' -1) = node k' v' < (widen l < u tl) tr' -0
... | = | cons k' v' l < u (tr' -0) = node k' v' = (widen l < u tl) tr' -0
... | > | cons k' v' l < u (tr' -1) = node k' v' = (widen l < u tl) tr' -1
... | > | cons k' v' l < u (tr' -0) = node k' v' > (widen l < u tl) tr' -0
delete k (node k1 v b tl tr) | tri > _ _ _ with delete k tr | b
... | tr' -1 | < = deleted (rotr k1 v tl tr')
... | tr' -1 | = = node k1 v < tl tr' -0
... | tr' -1 | > = node k1 v = tl tr' -1
... | tr' -0 | _ = node k1 v b tl tr' -0

```

## 9 Packaging

Users don't need to be exposed to the indices on the full tree type: here, we package it in three forms.

## 9.1 Dependent Map

```

module DependantMap where
data Map {v} (V : Key → Set v) : Set (k ⊔ v ⊔ r) where
  tree : ∀ {h} → Bounded.Tree V Bounded.⊥ Bounded.⊤ h → Map V

insertWith : ∀ {v} {V : Key → Set v} (k : Key)
  → V k
  → (V k → V k → V k)
  → Map V
  → Map V
insertWith k v f (tree tr) =
  tree (proj₂ (Bounded.insert k v f tr (lift tt , lift tt)))

insert : ∀ {v} {V : Key → Set v} (k : Key) → V k → Map V → Map V
insert k v = insertWith k v const

lookup : (k : Key) → ∀ {v} {V : Key → Set v} → Map V → Maybe (V k)
lookup k (tree tr) = Bounded.lookup k tr

delete : (k : Key) → ∀ {v} {V : Key → Set v} → Map V → Map V
delete k (tree tr) with Bounded.delete k tr
... | tr' Bounded.-0 = tree tr'
... | tr' Bounded.-1 = tree tr'

```

## 9.2 Non-Dependent (Simple) Map

```

module Map where
data Map {v} (V : Set v) : Set (k ⊔ v ⊔ r) where
  tree : ∀ {h} → Bounded.Tree (const V) Bounded.⊥ Bounded.⊤ h → Map V

insertWith : ∀ {v} {V : Set v} (k : Key) → V → (V → V → V) → Map V → Map V
insertWith k v f (tree tr) =
  tree (proj₂ (Bounded.insert k v f tr (lift tt , lift tt)))

insert : ∀ {v} {V : Set v} (k : Key) → V → Map V → Map V
insert k v = insertWith k v const

lookup : (k : Key) → ∀ {v} {V : Set v} → Map V → Maybe V
lookup k (tree tr) = Bounded.lookup k tr

delete : (k : Key) → ∀ {v} {V : Set v} → Map V → Map V
delete k (tree tr) with Bounded.delete k tr
... | tr' Bounded.-0 = tree tr'
... | tr' Bounded.-1 = tree tr'

```

### 9.3 Set

Note that we can't call the type itself "Set", as that's a reserved word in Agda.

```
module Sets where
data ⟨Set⟩ : Set (k ⊔ r) where
  tree : ∀ {h} → Bounded.Tree (const 1) Bounded.⊥ Bounded.⊤ h → ⟨Set⟩

insert : Key → ⟨Set⟩ → ⟨Set⟩
insert k (tree tr) =
  tree (proj₂ (Bounded.insert k tt const tr (lift tt , lift tt)))

member : Key → ⟨Set⟩ → Bool
member k (tree tr) = is-just (Bounded.lookup k tr)

delete : (k : Key) → ⟨Set⟩ → ⟨Set⟩
delete k (tree tr) with Bounded.delete k tr
... | tr' Bounded.-0 = tree tr'
... | tr' Bounded.-1 = tree tr'
```

## References

- [1] N. A. Danielsson, "The Agda standard library." [Online]. Available: <https://agda.github.io/agda-stdlib/README.html>
- [2] C. T. McBride, "How to Keep Your Neighbours in Order," in *Proceedings of the 19th ACM SIG-PLAN International Conference on Functional Programming*, ser. ICFP '14. ACM, pp. 297–309. [Online]. Available: <https://personal.cis.strath.ac.uk/conor.mcbride/pub/Pivotal.pdf>