

AVL Trees

D Oisín Kidney

October 24, 2018

Abstract

This is a verified implementation of AVL trees in Agda, taking ideas primarily from Conor McBride’s paper “How to Keep Your Neighbours in Order” [2] and the Agda standard library [1].

Contents

1	Introduction	2
2	Values	2
3	Bounded	2
4	Balance	3
5	The Tree Type	4
6	Rotations	4
6.1	Right Rotation	5
6.2	Left Rotation	6
7	Insertion	6
8	Lookup	7
9	Deletion	8
9.1	Uncons	8
9.2	Widening and Transitivity	9
9.3	Joining	9
9.4	Full Deletion	10
10	Alteration	10
11	Packaging	13
11.1	Dependent Map	13
11.2	Non-Dependent (Simple) Map	14
11.3	Set	15

1 Introduction

First, some imports.

```
{-# OPTIONS --without-K #-}

open import Relation.Binary
open import Level as Level using (Lift; lift; _⊔_; lower)
open import Data.Nat as ℕ using (ℕ; suc; zero; pred)
open import Data.Product
open import Data.Unit
open import Data.Maybe
open import Function
open import Data.Bool
open import Data.Empty
```

Next, we declare a module: the entirety of the following code is parameterized over the *key* type, and a strict total order on that key.

```
module AVL
  {k e r} (Key : Set k)
  {_ ≈ _ : Rel Key e}
  {_ < _ : Rel Key r}
  (isStrictTotalOrder : IsStrictTotalOrder _ ≈ _ < _)
  where

  open IsStrictTotalOrder isStrictTotalOrder
  open IsEquivalence isEquivalence
```

2 Values

Because we're not using propositional equality, we have to explicitly state that we want the equivalence relation to be substitutive for the value predicate.

```
record Value v : Set (e ⊔ Level.suc v ⊔ k) where
  field
    family : Key → Set v
    respects : family Respects _ ≈ _
open Value

Const : ∀ {v} (V : Set v) → Value v
family (Const V) _ = V
respects (Const V) _ x = x
```

3 Bounded

The basic idea of the verified implementation is to store in each leaf a proof that the upper and lower bounds of the trees to its left and right are ordered appropriately.

Accordingly, the tree type itself will have to have the upper and lower bounds in its indices. But what are the upper and lower bounds of a tree with no neighbours? To describe this case, we add lower and upper bounds to our key type.

```
module Bounded where

infix 5 [ ]

data [•] : Set k where
  [ ] [ ] : [•]
  [ ] : (k : Key) → [•]
```

This type itself admits an ordering relation.

```
infix 4 _<[_]_

_<[_]_ : [•] → [•] → Set r
[ ] <[_] [ ] = Lift r ⊥
[ ] <[_] [ ] = Lift r ⊤
[ ] <[_] [ _ ] = Lift r ⊤
[ ] <[_] [ ] = Lift r ⊥
[ _ ] <[_] [ ] = Lift r ⊥
[ _ ] <[_] [ ] = Lift r ⊤
[ x ] <[_] [ y ] = x < y
```

Finally, we can describe a value as being “in bounds” like so.

```
infix 4 _<_<_<_

_<_<_<_ : [•] → Key → [•] → Set r
l < x < u = l <[_] [ x ] × [ x ] <[_] u
```

4 Balance

To describe the balance of the tree, we use the following type:

```
data { _ ⊔ _ } ≡ _ : ℕ → ℕ → ℕ → Set where
  1+0 : ∀ {n} → ( suc n ⊔ n ) ≡ suc n
  0+0 : ∀ {n} → ( n ⊔ n ) ≡ n
  0+1 : ∀ {n} → ( n ⊔ suc n ) ≡ suc n
```

The tree can be either left- or right-heavy (by one), or even. The indices of the type are phrased as a proof:

$$\max(x, y) = z \quad (1)$$

The height of a tree is the maximum height of its two subtrees, plus one. Storing a proof of the maximum in this way will prove useful later.

We will also need some combinators for balance:

```

 $\sqcup\text{-zero}^r : \forall \{x\ y\ z\} \rightarrow \langle x \sqcup y \rangle \equiv z \rightarrow \langle z \sqcup x \rangle \equiv z$ 
 $\sqcup\text{-zero}^r\ 1+0 = 0+0$ 
 $\sqcup\text{-zero}^r\ 0+0 = 0+0$ 
 $\sqcup\text{-zero}^r\ 0+1 = 1+0$ 

 $\sqcup\text{-zero}^l : \forall \{x\ y\ z\} \rightarrow \langle x \sqcup y \rangle \equiv z \rightarrow \langle y \sqcup z \rangle \equiv z$ 
 $\sqcup\text{-zero}^l\ 1+0 = 0+1$ 
 $\sqcup\text{-zero}^l\ 0+0 = 0+0$ 
 $\sqcup\text{-zero}^l\ 0+1 = 0+0$ 

```

5 The Tree Type

The type itself is indexed by the lower and upper bounds, some value to store with the keys, and a height. In using the balance type defined earlier, we ensure that the children of a node cannot differ in height by more than 1. The bounds proofs also ensure that the tree must be ordered correctly.

```

data Tree {v}
  (V : Value v)
  (l u : [•]) : ℕ →
    Set (k ⊔ v ⊔ r) where
  leaf : (l < u : l [ < ] u) → Tree V l u 0
  node : ∀ {h lh rh}
    (k : Key)
    (v : V.family k)
    (bl : ⟨ lh ⊔ rh ⟩ ≡ h)
    (lk : Tree V l [ k ] lh)
    (ku : Tree V [ k ] u rh) →
    Tree V l u (suc h)

```

6 Rotations

AVL trees are rebalanced by rotations: if, after an insert or deletion, the balance invariant has been violated, one of these rotations is performed as correction.

Before we implement the rotations, we need a way to describe a tree which may have increased in height. We can do this with a *descriptive* type:

```

_1?+_⟨_⟩ : ∀ {ℓ} (T : ℕ → Set ℓ) → ℕ → Set ℓ
T 1?+_⟨ n ⟩ = ∃ [ inc? ] T (if inc? then suc n else n)

pattern 0+_ tr = false , tr
pattern 1+_ tr = true , tr

```

Later, we will also need to describe a tree which may have decreased in height. For this, we will use a *prescriptive* type (in other words, where the previous type was parameterized, this one will be indexed).

```
data _⟨_⟩?-1 {ℓ} (T : ℕ → Set ℓ) : ℕ → Set ℓ where
  -0 : ∀ {n} → T n → T ⟨ n ⟩?-1
  -1 : ∀ {n} → T n → T ⟨ suc n ⟩?-1
```

Whereas the previous construction would tell you the height of a tree after pattern matching on it, this definition will *refine* any information you already have about the height of the tree.

In certain circumstances, you can convert between the two:

```
1?+⟨_⟩⇒?-1 : ∀ {n ℓ} {T : ℕ → Set ℓ}
  → T 1?+⟨ n ⟩
  → T ⟨ suc n ⟩?-1
1?+⟨ 0+ x ⟩⇒?-1 = x -1
1?+⟨ 1+ x ⟩⇒?-1 = x -0
```

6.1 Right Rotation

When the left subtree becomes too heavy, we rotate the tree to the right.

```
rotr : ∀ {lb ub rh v} {V : Value v}
  → (k : Key)
  → V .family k
  → Tree V lb [ k ] (suc (suc rh))
  → Tree V [ k ] ub rh
  → Tree V lb ub 1?+⟨ suc (suc rh) ⟩
```

This rotation comes in two varieties: single and double. Single rotation can be seen in figure 1.




Figure 1: Single right-rotation

```
rotr x xv (node y yv 1+0 a b) c =
  0+ (node y yv 0+0 a (node x xv 0+0 b c))
rotr x xv (node y yv 0+0 a b) c =
  1+ (node y yv 0+1 a (node x xv 1+0 b c))
```

And double rotation in figure 2.

```
rotr x xv (node y yv 0+1 a (node z zv bl b c)) d =
  0+ (node z zv 0+0 (node y yv (⊔-zeror bl) a b) (node x xv (⊔-zerol bl) c d))
```




Figure 2: Double right-rotation

6.2 Left Rotation

Left-rotation is essentially the inverse of right.

```

rotl : ∀ {lb ub lh v} {V : Value v}
  → (k : Key)
  → V.family k
  → Tree V lb [ k ] lh
  → Tree V [ k ] ub (suc (suc lh))
  → Tree V lb ub 1?+ ( suc (suc lh) )

```

Single (seen in figure 3).




Figure 3: Single left-rotation

```

rotl x xv c (node y yv 0+1 b a) =
  0+ (node y yv 0+0 (node x xv 0+0 c b) a)
rotl x xv c (node y yv 0+0 b a) =
  1+ (node y yv 1+0 (node x xv 0+1 c b) a)

```

and double (figure 4):

```

rotl x xv d (node y yv 1+0 (node z zv bl c b) a) =
  0+ (node z zv 0+0 (node x xv (L-zeror bl) d c) (node y yv (L-zerol bl) b a))

```

7 Insertion

After the rotations, insertion is relatively easy. We allow the caller to supply a combining function.




Figure 4: Double left-rotation

```

insert : ∀ {l u h v} {V : Value v} (k : Key)
  → V.family k
  → (V.family k → V.family k → V.family k)
  → Tree V l u h
  → l < k < u
  → Tree V l u 1?+⟨ h ⟩
insert v vc f (leaf l < u) (l, u) = 1+ (node v vc 0+0 (leaf l) (leaf u))
insert v vc f (node k kc bl tl tr) prf with compare v k
insert v vc f (node k kc bl tl tr) (l, _)
  | tri < a _ _ with insert v vc f tl (l, a)
... | 0+ tl' = 0+ (node k kc bl tl' tr)
... | 1+ tl' with bl
... | 1+0 = rotr k kc tl' tr
... | 0+0 = 1+ (node k kc 1+0 tl' tr)
... | 0+1 = 0+ (node k kc 0+0 tl' tr)
insert {V = V} v vc f (node k kc bl tl tr) _ v ≈ k (f vc (respects V (sym v ≈ k) kc)) bl tl tr)
insert v vc f (node k kc bl tl tr) (_, u)
  | tri > _ _ c with insert v vc f tr (c, u)
... | 0+ tr' = 0+ (node k kc bl tl tr')
... | 1+ tr' with bl
... | 1+0 = 0+ (node k kc 0+0 tl tr')
... | 0+0 = 1+ (node k kc 0+1 tl tr')
... | 0+1 = rotl k kc tl tr'

```

8 Lookup

Lookup is also very simple. No invariants are needed here.

```

lookup : (k : Key)
  → ∀ {l u s v} {V : Value v}
  → Tree V l u s
  → Maybe (V.family k)
lookup k (leaf l < u) = nothing
lookup k {V = V} (node v vc _ tl tr) with compare k v

```

```

... | tri< _ _ _ = lookup k tl
... | tri≈ _ k≈v _ = just (respects V (sym k≈v) vc)
... | tri> _ _ _ = lookup k tr

```

9 Deletion

Deletion is by far the most complex operation out of the three provided here. For deletion from a normal BST, you go to the node where the desired value is, perform an “uncons” operation on the right subtree, use that as your root node, and merge the two remaining children.

9.1 Uncons

First then, we need to define “uncons”. We’ll use a custom type as the return type from our uncons function, which stores the minimum element from the tree, and the rest of the tree:

```

record Cons {v}
  (V : Value v)
  (lb ub : [•])
  (h : ℕ) : Set (k ⊔ v ⊔ r) where
constructor cons
field
  head : Key
  val  : family V head
  l<u  : lb [<] [ head ]
  tail : Tree V [ head ] ub 1?+⟨ h ⟩

```

You’ll notice it also stores a proof that the extracted element preserves the lower bound.

```

uncons : ∀ {lb ub h lh rh v} {V : Value v}
  → (k : Key)
  → V.family k
  → ⟨ lh ⊔ rh ⟩≡ h
  → Tree V lb [ k ] lh
  → Tree V [ k ] ub rh
  → Cons V lb ub h
uncons k v b (leaf l<u) tr = cons k v l<u (case b of
  λ { 0+1 → 0+ tr
    ; 0+0 → 0+ tr })
uncons k v b (node kl vl bl tl trl) tr with uncons kl vl bl tl trl
... | cons k' v' l<u tail = cons k' v' l<u (case tail of
  λ { (1+ tl') → 1+ (node k v b tl' tr)
    ; (0+ tl') → case b of
      λ { 1+0 → 0+ node k v 0+0 tl' tr
        ; 0+0 → 1+ node k v 0+1 tl' tr
        ; 0+1 → rotl k v tl' tr }}})

```


9.2 Widening and Transitivity

To join the two subtrees together after a deletion operation, we need to weaken (or ext) the bounds of the left tree. This is an $\mathcal{O}(\log n)$ operation.

For the exting, we'll need some properties on orderings:

$$\begin{aligned}
& \text{x}\<[] : \forall \{x\} \rightarrow x \text{ [<] } [] \rightarrow \text{Lift } r \perp \\
& \text{x}\<[] \{[]\} = \text{lift} \circ \text{lower} \\
& \text{x}\<[] \{[]\} = \text{lift} \circ \text{lower} \\
& \text{x}\<[] \{[_]\} = \text{lift} \circ \text{lower} \\
& \text{[<]-trans} : \forall x \{y \ z\} \rightarrow x \text{ [<] } y \rightarrow y \text{ [<] } z \rightarrow x \text{ [<] } z \\
& \text{[<]-trans} [] \{y\} \{[]\} _ \quad y < z = \text{x}\<[] \{x = y\} y < z \\
& \text{[<]-trans} [] \{ _ \} \{[]\} _ \quad _ = _ \\
& \text{[<]-trans} [] \{ _ \} \{[_]\} _ \quad _ = _ \\
& \text{[<]-trans} [] \{ _ \} \{ _ \} \text{(lift ())} _ \quad _ = _ \\
& \text{[<]-trans} [_] \{y\} \{[]\} _ \quad y < z = \text{x}\<[] \{x = y\} y < z \\
& \text{[<]-trans} [_] \{ _ \} \{[]\} _ \quad _ = _ \\
& \text{[<]-trans} [_] \{[]\} \{[_]\} \text{(lift ())} _ \quad _ = _ \\
& \text{[<]-trans} [_] \{[]\} \{[_]\} _ \text{(lift ())} \quad _ = _ \\
& \text{[<]-trans} [x] \{[y]\} \{[z]\} \quad x < y \ y < z = \\
& \text{IsStrictTotalOrder.trans isStrictTotalOrder } x < y \ y < z
\end{aligned}$$

Finally, the ext function itself simply walks down the right branch of the tree until it hits a leaf.

$$\begin{aligned}
& \text{ext} : \forall \{lb \ ub \ ub' \ h \ v\} \{V : \text{Value } v\} \\
& \quad \rightarrow ub \text{ [<] } ub' \\
& \quad \rightarrow \text{Tree } V \ lb \ ub \ h \\
& \quad \rightarrow \text{Tree } V \ lb \ ub' \ h \\
& \text{ext } \{lb\} \ ub < ub' \ (\text{leaf } l < u) = \text{leaf } ([<]-trans \ lb \ l < u \ ub < ub') \\
& \text{ext } ub < ub' \ (\text{node } k \ v \ bl \ tl \ tr) = \text{node } k \ v \ bl \ tl \ (\text{ext } ub < ub' \ tr)
\end{aligned}$$

9.3 Joining

Once we have the two subtrees that will form the children of our replaced node, we need to join them together, adjusting the types accordingly.

$$\begin{aligned}
& \text{join} : \forall \{lb \ ub \ lh \ rh \ h \ v \ k\} \{V : \text{Value } v\} \\
& \quad \rightarrow \text{Tree } V \ [k] \ ub \ rh \\
& \quad \rightarrow \langle \ lh \sqcup \ rh \rangle \equiv h \\
& \quad \rightarrow \text{Tree } V \ lb \ [k] \ lh \\
& \quad \rightarrow \text{Tree } V \ lb \ ub \ 1?+ \langle h \rangle \\
& \text{join } (\text{leaf } k < ub) \ 1+0 \ tl = 0+ \text{ext } k < ub \ tl \\
& \text{join } \{lb\} \ (\text{leaf } k < ub) \ 0+0 \ (\text{leaf } lb < k) = \\
& \quad 0+ \text{leaf } ([<]-trans \ lb \ lb < k \ k < ub) \\
& \text{join } (\text{node } k_r \ v_r \ b_r \ tl_r \ tr_r) \ b \ tl \ \text{with } \text{uncons } k_r \ v_r \ b_r \ tl_r \ tr_r \\
& \dots \mid \text{cons } k' \ v' \ l < u \ (1+ \ tr') = 1+ \ \text{node } k' \ v' \ b \ (\text{ext } l < u \ tl) \ tr' \\
& \dots \mid \text{cons } k' \ v' \ l < u \ (0+ \ tr') \ \text{with } b \\
& \dots \mid 1+0 = \text{rot}^r \ k' \ v' \ (\text{ext } l < u \ tl) \ tr'
\end{aligned}$$

```

... | 0+0 = 1+ node  $k' v' 1+0$  (ext  $l < u$   $tl$ )  $tr'$ 
... | 0+1 = 0+ node  $k' v' 0+0$  (ext  $l < u$   $tl$ )  $tr'$ 

```

9.4 Full Deletion

The deletion function is by no means simple, but it does maintain the correct complexity bounds.

```

delete :  $\forall \{lb \ ub \ h \ v\} \{V : \text{Value } v\}$ 
         $\rightarrow (k : \text{Key})$ 
         $\rightarrow \text{Tree } V \ lb \ ub \ h$ 
         $\rightarrow \text{Tree } V \ lb \ ub \langle h \rangle ?-1$ 
delete  $x$  (leaf  $l < u$ ) = leaf  $l < u -0$ 
delete  $x$  (node  $y \ yv \ b \ l \ r$ ) with compare  $x \ y$ 
delete  $x$  (node  $y \ yv \ b \ l \ r$ ) | tri  $\approx$   $\_ \_ \_$  =  $1?+\langle \text{join } r \ b \ l \rangle \Rightarrow ?-1$ 
delete  $x$  (node  $y \ yv \ b \ l \ r$ ) | tri  $<$   $a \_ \_$  with delete  $x \ l$ 
... |  $l' -0$  = node  $y \ yv \ b \ l' \ r -0$ 
... |  $l' -1$  with  $b$ 
... |  $1+0$  = node  $y \ yv \ 0+0 \ l' \ r -1$ 
... |  $0+0$  = node  $y \ yv \ 0+1 \ l' \ r -0$ 
... |  $0+1$  =  $1?+\langle \text{rot}^l \ y \ yv \ l' \ r \rangle \Rightarrow ?-1$ 
delete  $x$  (node  $y \ yv \ b \ l \ r$ ) | tri  $>$   $\_ \_ c$  with delete  $x \ r$ 
... |  $r' -0$  = node  $y \ yv \ b \ l \ r' -0$ 
... |  $r' -1$  with  $b$ 
... |  $1+0$  =  $1?+\langle \text{rot}^r \ y \ yv \ l \ r' \rangle \Rightarrow ?-1$ 
... |  $0+0$  = node  $y \ yv \ 1+0 \ l \ r' -0$ 
... |  $0+1$  = node  $y \ yv \ 0+0 \ l \ r' -1$ 

```

10 Alteration

This is a combination of insertion and deletion: it lets the user supply a function to modify, insert, or remove an element, depending on the element already in the tree.

As it can both increase and decrease the size of the tree, we need a wrapper to represent that:

```

data  $\_ \langle \_ \rangle \pm 1 \{ \ell \} (T : \mathbb{N} \rightarrow \text{Set } \ell) : \mathbb{N} \rightarrow \text{Set } \ell$  where
   $1+\langle \_ \rangle : \forall \{n\} \rightarrow T (\text{suc } n) \rightarrow T \langle n \rangle \pm 1$ 
   $\langle \_ \rangle : \forall \{n\} \rightarrow T \ n \rightarrow T \langle n \rangle \pm 1$ 
   $\langle \_ \rangle -1 : \forall \{n\} \rightarrow T \ n \rightarrow T \langle \text{suc } n \rangle \pm 1$ 

 $1?+\langle \_ \rangle \Rightarrow -1 : \forall \{n \ell\} \{T : \mathbb{N} \rightarrow \text{Set } \ell\}$ 
   $\rightarrow T \ 1?+\langle n \rangle$ 
   $\rightarrow T \langle \text{suc } n \rangle \pm 1$ 
 $1?+\langle 0+ \ x \rangle \Rightarrow -1 = \langle x \rangle -1$ 
 $1?+\langle 1+ \ x \rangle \Rightarrow -1 = \langle x \rangle$ 

```

```

1?+⟨_⟩⇒+1 : ∀ {n ℓ} {T : ℕ → Set ℓ}
  → T 1?+⟨ n ⟩
  → T ⟨ n ⟩±1
1?+⟨ 0+ x ⟩⇒+1 = ⟨ x ⟩
1?+⟨ 1+ x ⟩⇒+1 = 1+⟨ x ⟩

```

And then the function itself. It's long, but you should be able to see the deletion and insertion components.

```

alter : ∀ {lb ub h v} {V : Value v}
  → (k : Key)
  → (Maybe (V.family k) → Maybe (V.family k))
  → Tree V lb ub h
  → lb < k < ub
  → Tree V lb ub ⟨ h ⟩±1
alter x f (leaf l<u) (l , u) with f nothing
... | just xv = 1+⟨ node x xv 0+0 (leaf l) (leaf u) ⟩
... | nothing = ⟨ leaf l<u ⟩
alter x f (node y yv b tl tr) (l , u)
  with compare x y
alter {V = V} x f (node y yv b tl tr) (l , u)
  | tri≈ _ x≈y _ with f (just (respects V (sym x≈y) yv))
... | just xv = ⟨ node y (respects V x≈y xv) b tl tr ⟩
... | nothing = 1?+⟨ join tr b tl ⟩⇒-1
alter x f (node y yv b tl tr) (l , u)
  | tri< a _ _ with alter x f tl (l , a) | b
... | ⟨ tl' ⟩ | _ = ⟨ node y yv b tl' tr ⟩
... | 1+⟨ tl' ⟩ | 1+0 = 1?+⟨ rotr y yv tl' tr ⟩⇒+1
... | 1+⟨ tl' ⟩ | 0+0 = 1+⟨ node y yv 1+0 tl' tr ⟩
... | 1+⟨ tl' ⟩ | 0+1 = ⟨ node y yv 0+0 tl' tr ⟩
... | ⟨ tl' ⟩-1 | 1+0 = ⟨ node y yv 0+0 tl' tr ⟩-1
... | ⟨ tl' ⟩-1 | 0+0 = ⟨ node y yv 0+1 tl' tr ⟩
... | ⟨ tl' ⟩-1 | 0+1 = 1?+⟨ rotl y yv tl' tr ⟩⇒-1
alter x f (node y yv b tl tr) (l , u)
  | tri> _ _ c with alter x f tr (c , u) | b
... | ⟨ tr' ⟩ | _ = ⟨ node y yv b tl tr' ⟩
... | 1+⟨ tr' ⟩ | 1+0 = ⟨ node y yv 0+0 tl tr' ⟩
... | 1+⟨ tr' ⟩ | 0+0 = 1+⟨ node y yv 0+1 tl tr' ⟩
... | 1+⟨ tr' ⟩ | 0+1 = 1?+⟨ rotl y yv tl tr' ⟩⇒+1
... | ⟨ tr' ⟩-1 | 1+0 = 1?+⟨ rotr y yv tl tr' ⟩⇒-1
... | ⟨ tr' ⟩-1 | 0+0 = ⟨ node y yv 1+0 tl tr' ⟩
... | ⟨ tr' ⟩-1 | 0+1 = ⟨ node y yv 0+0 tl tr' ⟩-1

```

We can also write alterF, in the lens style.

```

open import Category.Functor using (RawFunctor)

MaybeVal : ∀ {v} (V : Set v) → Set (k ⊔ r ⊔ v)
MaybeVal V = Lift (k ⊔ r) (Maybe V)

```

```

alterF : ∀ {lb ub h v} {V : Value v} {R : Set (k ⊔ r ⊔ v)}
  → (Tree V lb ub ⟨ h ⟩ ±1 → R)
  → (x : Key)
  → ∀ {F : Set (k ⊔ r ⊔ v) → Set (k ⊔ r ⊔ v)}
    { {functor : RawFunctor F} }
  → (Maybe (family V x) → F (MaybeVal (family V x)))
  → Tree V lb ub h
  → lb < x < ub
  → F R

alterF {lb} {ub} {h} {_} {V} {R} kl x {F} { {functor} } f root bnds
= go root bnds id
where
  _<&>_ : ∀ {A B} → F A → (A → B) → F B
  xs <&> f = RawFunctor._<$>_ functor f xs
  go : ∀ {lbl ubl hl}
    → Tree V lbl ubl hl
    → lbl < x < ubl
    → (Tree V lbl ubl ⟨ hl ⟩ ±1 → Tree V lb ub ⟨ h ⟩ ±1)
    → F R
  go (leaf l<u) (l , u) k = f nothing <&>
    λ { (lift nothing) → kl ⟨ root ⟩
      ; (lift (just xv)) → kl (k 1+⟨ node x xv 0+0 (leaf l) (leaf u) ⟩) }
  go (node y yv b tl tr) (l , u) k with compare x y
  go (node y yv b tl tr) (l , u) k | tri≈ _ x≈y _ = f (just (respects V (sym x≈y) yv)) <&>
    λ { (lift nothing) → kl (k 1?+⟨ join tr b tl ⟩ ⇒ -1)
      ; (lift (just xv)) → kl (k ⟨ node y (respects V x≈y xv) b tl tr ⟩) }
  go (node y yv b tl tr) (l , u) k | tri< a _ _ = go tl (l , a) (k ∘
    λ { ⟨ tll ⟩ → ⟨ node y yv b tll tr ⟩
      ; 1+⟨ tll ⟩ → case b of
        λ { 1+0 → 1?+⟨ rotr y yv tll tr ⟩ ⇒ +1
          ; 0+0 → 1+⟨ node y yv 1+0 tll tr ⟩
          ; 0+1 → ⟨ node y yv 0+0 tll tr ⟩ }
      ; ⟨ tll ⟩ -1 → case b of
        λ { 1+0 → ⟨ node y yv 0+0 tll tr ⟩ -1
          ; 0+0 → ⟨ node y yv 0+1 tll tr ⟩
          ; 0+1 → 1?+⟨ rotl y yv tll tr ⟩ ⇒ -1 } })
  go (node y yv b tl tr) (l , u) k | tri> _ _ c = go tr (c , u) (k ∘
    λ { ⟨ trl ⟩ → ⟨ node y yv b tl trl ⟩
      ; 1+⟨ trl ⟩ → case b of
        λ { 1+0 → ⟨ node y yv 0+0 tl trl ⟩
          ; 0+0 → 1+⟨ node y yv 0+1 tl trl ⟩
          ; 0+1 → 1?+⟨ rotl y yv tl trl ⟩ ⇒ +1 }
      ; ⟨ trl ⟩ -1 → case b of
        λ { 1+0 → 1?+⟨ rotr y yv tl trl ⟩ ⇒ -1
          ; 0+0 → ⟨ node y yv 1+0 tl trl ⟩

```

; 0+1 → ⟨ node y yv 0+0 tl tr^l -1 ⟩ } }

11 Packaging

Users don't need to be exposed to the indices on the full tree type: here, we package it in three forms.

11.1 Dependent Map

```

module DependantMap where
data Map {v} (V : Value v) : Set (k ⊔ v ⊔ r) where
  tree : ∀ {h}
    → Bounded.Tree V Bounded.[ ] Bounded.[ ] h
    → Map V

insertWith : ∀ {v} {V : Value v} (k : Key)
  → V.family k
  → (V.family k → V.family k → V.family k)
  → Map V
  → Map V
insertWith k v f (tree tr) =
  tree (proj₂ (Bounded.insert k v f tr (lift tt , lift tt)))

insert : ∀ {v}
  {V : Value v}
  (k : Key) →
  V.family k →
  Map V →
  Map V
insert k v = insertWith k v const

empty : ∀ {v} {V : Value v} → Map V
empty = tree (Bounded.leaf (lift tt))

lookup : (k : Key)
  → ∀ {v} {V : Value v}
  → Map V
  → Maybe (V.family k)
lookup k (tree tr) = Bounded.lookup k tr

delete : (k : Key)
  → ∀ {v} {V : Value v}
  → Map V
  → Map V
delete k (tree tr) with Bounded.delete k tr
... | trl Bounded.-0 = tree trl
... | trl Bounded.-1 = tree trl

```

```

alter : (k : Key)
  → ∀ {v} {V : Value v}
  → (Maybe (V.family k) → Maybe (V.family k))
  → Map V
  → Map V
alter k f (tree tr) with Bounded.alter k f tr (lift tt , lift tt)
... | Bounded.1+⟨ tr' ⟩ = tree tr'
... | Bounded.⟨ tr' ⟩    = tree tr'
... | Bounded.⟨ tr' ⟩-1 = tree tr'

```

11.2 Non-Dependent (Simple) Map

```

module Map where
data Map {v} (V : Set v) : Set (k ⊔ v ⊔ r) where
  tree : ∀ {h}
    → Bounded.Tree (Const V) Bounded.[ ] Bounded.[ ] h
    → Map V

insertWith : ∀ {v} {V : Set v} (k : Key)
  → V
  → (V → V → V)
  → Map V
  → Map V
insertWith k v f (tree tr) =
  tree (proj₂ (Bounded.insert k v f tr (lift tt , lift tt)))

empty : ∀ {v} {V : Set v} → Map V
empty = tree (Bounded.leaf (lift tt))

insert : ∀ {v} {V : Set v} (k : Key) → V → Map V → Map V
insert k v = insertWith k v const

lookup : (k : Key) → ∀ {v} {V : Set v} → Map V → Maybe V
lookup k (tree tr) = Bounded.lookup k tr

delete : (k : Key) → ∀ {v} {V : Set v} → Map V → Map V
delete k (tree tr) with Bounded.delete k tr
... | tr' Bounded.-0 = tree tr'
... | tr' Bounded.-1 = tree tr'

alter : (k : Key)
  → ∀ {v} {V : Set v}
  → (Maybe V → Maybe V)
  → Map V
  → Map V
alter k f (tree tr) with Bounded.alter k f tr (lift tt , lift tt)
... | Bounded.1+⟨ tr' ⟩ = tree tr'
... | Bounded.⟨ tr' ⟩    = tree tr'

```

```
... | Bounded.< tr' >-1 = tree tr'
```

11.3 Set

Note that we can't call the type itself "Set", as that's a reserved word in Agda.

```
module Sets where
data <Set> : Set (k ⊔ r) where
  tree : ∀ {h}
    → Bounded.Tree (Const T) Bounded.[] Bounded.[] h
    → <Set>

insert : Key → <Set> → <Set>
insert k (tree tr) =
  tree (proj2 (Bounded.insert k tt const tr (lift tt , lift tt)))

member : Key → <Set> → Bool
member k (tree tr) = is-just (Bounded.lookup k tr)

delete : (k : Key) → <Set> → <Set>
delete k (tree tr) with Bounded.delete k tr
... | tr' Bounded.-0 = tree tr'
... | tr' Bounded.-1 = tree tr'
```

References

- [1] N. A. Danielsson, "The Agda standard library." [Online]. Available: <https://agda.github.io/agda-stdlib/README.html>
- [2] C. T. McBride, "How to Keep Your Neighbours in Order," in *Proceedings of the 19th ACM SIG-PLAN International Conference on Functional Programming*, ser. ICFP '14. ACM, pp. 297–309. [Online]. Available: <https://personal.cis.strath.ac.uk/conor.mcbride/pub/Pivotal.pdf>